Asymptotes verticales et horizontales

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. …Ça fait un bail !

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,setoptions):
  setoptions(size=[300,300],axesfont=[times,roman,8],color=navy ):
```

Étude de la fonction f définie par $f(x) = \dfrac{3\,x^2+1}{5\,x^2-7}$

Déterminons en tout premier lieu le domaine de la fonction f. Le domaine de cette fonction rationnelle est dom $f = \mathbb{R} \setminus \{x \in \mathbb{R} \mid 5\,x^2 - 7 = 0\}$. Obtenons alors les nombres réels qui annulent $5\,x^2 - 7$.

```
> f:=x->(3*x^2+1)/(5*x^2-7);
  Racines:=solve(denom(f(x))=0,x);
```

$$f := x \mapsto \frac{3 \cdot x^2 + 1}{5 \cdot x^2 - 7}$$

$$Racines := \frac{\sqrt{35}}{5}, -\frac{\sqrt{35}}{5} \qquad \textbf{(2.1)}$$

Donc, le dom f $= \mathbb{R} \setminus \left\{ \frac{\sqrt{35}}{5}, -\frac{\sqrt{35}}{5} \right\}$. Ainsi, les candidats à retenir pour analyser la discontinuité sont $x = \frac{\sqrt{35}}{5}$ et $x = -\frac{\sqrt{35}}{5}$. Commençons notre analyse avec $x = \frac{\sqrt{35}}{5}$.

```
>  Limit(f(x),x=Racines[1])=limit(f(x),x=Racines[1]);
```

$$\lim_{x \to \frac{\sqrt{35}}{5}} \frac{3\,x^2+1}{5\,x^2-7} = undefined \qquad \textbf{(2.2)}$$

Nous devons considérer les limites directionnelles en $x = \frac{\sqrt{35}}{5}$. Évaluons alors la limite à gauche et la limite et la droite.

```
> Limit(f(x),x=Racines[1],left)=limit(f(x),x=Racines[1],left);
```

```
Limit(f(x),x=Racines[1],right)=limit(f(x),x=Racines[1],right);
```

$$\lim_{x \to \left(\frac{\sqrt{35}}{5}\right)^-} \frac{3x^2+1}{5x^2-7} = -\infty$$

$$\lim_{x \to \left(\frac{\sqrt{35}}{5}\right)^+} \frac{3x^2+1}{5x^2-7} = \infty \qquad \textbf{(2.3)}$$

La nature de la discontinuité en $x=\frac{\sqrt{35}}{5}$ est donc une discontinuité infinie. Il y a donc un comportement asymptotique de la fonction autour de la droite d'équation $x=\frac{\sqrt{35}}{5}$. Cette droite est appelée *asymptote verticale*.

Esquissons le graphique de la fonction pour des valeurs de x près de cette asymptote verticale d'équation $x=\frac{\sqrt{35}}{5}$: traçons donc la fonction sur l'intervalle $[0,75;\ 1,5]$. Superposons-y également l'asymptote verticale tracée en tirets de couleur khaki.

```
> Graphe_1:=plot([x,f(x),x=0.75..1.5],discont=true):
  Asymptote_V1:=plot([Racines[1],y,y=-6..6],linestyle=3,thickness=
  0,color=khaki):
  display({Graphe_1,Asymptote_V1},view=[-2..2,-6..6]);
```



Reste alors à déterminer la nature de la discontinuité en $x=-\frac{\sqrt{35}}{5}$.

```
> Limit(f(x),x=Racines[2])=limit(f(x),x=Racines[2]);
```

$$\lim_{x \to -\frac{\sqrt{35}}{5}} \frac{3x^2+1}{5x^2-7} = \mathit{undefined} \qquad \textbf{(2.4)}$$

Ici aussi, il est donc nécessaire d'évaluer à nouveau la limite en évaluant les deux limites directionnelles.

```
> Limit(f(x),x=Racines[2],left)=limit(f(x),x=Racines[2],left);
  Limit(f(x),x=Racines[2],right)=limit(f(x),x=Racines[2],right);
```

$$\lim_{x \to \left(-\frac{\sqrt{35}}{5}\right)^-} \frac{3x^2+1}{5x^2-7} = \infty$$

(2.5)

$$\lim_{x \to \left(-\frac{\sqrt{35}}{5}\right)^+} \frac{3x^2+1}{5x^2-7} = -\infty \tag{2.5}$$

La nature de la discontinuité en $x = -\frac{\sqrt{35}}{5}$ est aussi une discontinuité infinie. Il y a donc un comportement asymptotique de f autour de la droite d'équation $x = -\frac{\sqrt{35}}{5}$ Cette droite est donc une *asymptote verticale.*
Esquissons le graphique de la fonction pour des valeurs de x près de cette asymptote verticale d'équation $x = -\frac{\sqrt{35}}{5}$: traçons f sur l'intervalle $[-1,5\,;-0,75]$. Superposons-y l'asymptote verticale tracée en tirets.

```
> Graphe_2:=plot([x,f(x),x=-1.5..-0.75],discont=true):
  Asymptote_V2:=plot([Racines[2],y,y=-6..6],linestyle=3,thickness=
  0,color=khaki):
  display({Graphe_2,Asymptote_V2},view=[-2..2,-6..6]);
```



Pour faire l'étude d'un éventuel comportement asymptotique horizontal de la fonction f, il nous faudra évaluer les limites à l'infini: soit $\lim_{x \to -\infty} f(x)$ et $\lim_{x \to \infty} f(x)$. Chacun de ces deux calculs de limite doit être pertinent: il faut s'assurer que la fonction f est toujours définie lorsque $x \to -\infty$ ou lorsque $x \to \infty$. C'est le cas en raison du domaine de la fonction f.

Évaluons d'abord la $\lim_{x \to -\infty} f(x)$.

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x \to -\infty} \frac{3x^2+1}{5x^2-7} = \frac{3}{5} \tag{2.6}$$

Lorsque $x \to -\infty$, $f(x) \to \frac{3}{5}$. Autrement dit, lorsque $x \to -\infty$, les images par la fonction f se rapprochent de plus en plus de la valeur $\frac{3}{5}$. Donc, graphiquement, les points de la fonction f se rapprocheront de plus en plus de la droite d'équation $y = \frac{3}{5}$ lorsque $x \to -\infty$. Afin de mieux illustrer graphiquement le comportement des images pour des valeurs de x de plus en plus petites (au sens algébrique), nous tracerons en tirets la droite horizontale d'équation

$y = \frac{3}{5}$. Cette droite est appelée *asymptote horizontale*. On dit alors que la fonction f a un comportement asymptotique horizontal dans la partie négative de l'abscisse.

Esquissons le graphique de la fonction f pour des valeurs de y proches de cette asymptote horizontale d'équation $y = \frac{3}{5}$. Il faut donc considérer des valeurs de x "assez petites mais pas trop" pour que leur image y soit visuellement assez près de $\frac{3}{5}$ et correctement visible sur le graphique. Un tracé pour $x \in [-4, -2]$ va faire l'affaire.

```
> Graphe_3:=plot([x,f(x),x=-4..-2]):
  Asymptote_H1:=plot([x,3/5,x=-4..0],linestyle=3,thickness=0,color=
  khaki):
  display({Graphe_3,Asymptote_H1},view=[-4..4,-6..6]);
```



Reste à évaluer la $\lim_{x \to \infty} \frac{3x^2+1}{5x^2-7}$.

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \frac{3x^2+1}{5x^2-7} = \frac{3}{5} \qquad \textbf{(2.7)}$$

Il y a aussi un comportement asymptotique horizontal de la fonction f dans la partie positive de l'abscisse. Les points de la fonction se rapprochent de plus en plus de la droite d'équation $y = \frac{3}{5}$.

```
> Graphe_4:=plot([x,f(x),x=2..4]):
  Asymptote_H2:=plot([x,3/5,x=0..4],linestyle=3,thickness=0,color=
  khaki):
  display({Graphe_4,Asymptote_H2\},view=[-4..4,-6..6]);
```



Puisque les asymptotes horizontales ont la même équation dans la partie positive et dans la partie négative de l'abscisse, on dira, simplement, qu'il y a, pour la fonction f, une asymptote horizontale d'équation $y = \frac{3}{5}$. On pourait avoir avec certaines fonctions un comportement asymptotique horizontal différent lorsque $x \to -\infty$ et $x \to \infty$. On pourrait même avec une asymptote horizontale lorsque $x \to -\infty$ et ne pas en avoir lorsque $x \to \infty$ ou vice et versa.

Superposons maintenant, dans un même graphique, l'ensemble des éléments que nous avons obtenus dans cette étude.

```
> display({Graphe_1,Graphe_2,Graphe_3,Graphe_4,
          Asymptote_V1,Asymptote_V2,Asymptote_H1,Asymptote_H2},
          view=[-4..4,-6..6]);
```



Avec cette étude, avons-nous une bonne idée du graphique complet de la fonction f ? C'est à voir !

Étude de la fonction g définie par $g(x) = \sqrt[3]{\frac{x(x-1)}{x^2-1}}$

Recherchons d'abord le domaine de la fonction g.

Le radicand d'une racine cubique peut être négatif ou non-négatif. Il suffit donc que le radicand soit défini. Alors, pour que la fraction $\frac{x(x-1)}{x^2-1}$ puisse l'être, il suffit que le dénominateur soit non nul. C'est le cas si $x \neq \pm 1$.
Donc le dom g $= \mathbb{R} \setminus \{\pm 1\}$.

Alors, les candidats à la discontinuité sont $x = -1$ et $x = 1$. Obtenons la liste des candidats avec la macro-commande `discont`, il vous faut donc créer la fonction g.
Rappelons qu'il y a, avec Maple, deux macro-commandes pour extraire la racine énième: `root` et `surd`. On peut aussi transposer le radical en puissance fractionnaire: $\sqrt[n]{(\ )}$ devenant $(\ )^{\frac{1}{3}}$, ce qui est reconnu par l'évluateur à utiliser `root`. Nous avons eu l'occasion (voir le document *Limite et continuité.mw*) de constater que le calcul de la limite dans $\mathbb{C}$ ne donne pas nécessairement la même réponse dans $\mathbb{R}$. Pour faire l'analyse, dans les réels, des candidats à la discontinuité d'une fonction réelle d'une variable réelle, la macro-commande `root` ou la notation puissance fractionnaire **est à éviter** : l'évaluateur opère par défaut, comme vous le savez, tous les calculs dans l'ensemble des nombres complexes $\mathbb{C}$.

Créons donc la fonction g avec la macro-commande `surd` et confirmons la liste des candidats à la discontinuité pour la fonction g avec la macro-commande `discont`.

```
> g:=x->surd(x*(x-1)/(x^2-1),3);
  discont(g(x),x)
```

$$g := x \mapsto \sqrt[3]{\frac{x \cdot (x-1)}{x^2-1}}$$

$$\{-1, 1\} \tag{3.1}$$

Analysons d'abord le candidat $x = 1$ en évaluant la limite de g en $x = 1$.

```
> Limit(g(x),x=1)=limit(g(x),x=1);
```

$$\lim_{x \to 1} \sqrt[3]{\frac{x\,(x-1)}{x^2-1}} = \frac{4^{1/3}}{2} \tag{3.2}$$

Puisque la $\lim_{x \to 1} g(x)$ existe, nous devons conclure qu'il y a en $x = 1$ une discontinuité non essentielle qui est un trou. Il n'y a donc pas d'asymptote verticale en $x = 1$.

Esquissons le graphique de la fonction g dans un voisinage de $x = 1$. Traçons le graphique de g sur l'intervalle $[0,5;\ 1,5]$.
Contrôlons l'affichage des axes avec l'option view=[-3..3,-4..4].

```
> Graphe_1:=plot([x,g(x),x=.5..1.5]):
  Trou:=plot([[1,1/2*surd(4,3)]],style=point,symbol=solidcircle,
  symbolsize=15,color=orange):
  display({Trou,Graphe_1},view=[-3..3,-4..4]);
```



Pour mieux visualiser le trou en $x = 1$, il faut "l'exagérer" comme nous l'avons fait précédemment en traçant un petit cercle de couleur orange au point $\left(1, \frac{\sqrt[3]{4}}{2}\right)$.

Analysons maintenant le candidat $x = -1$.

```
> Limit(g(x),x=-1)=limit(g(x),x=-1);
```

$$\lim_{x \to -1} \sqrt[3]{\frac{x\,(x-1)}{x^2-1}} = undefined \qquad \textbf{(3.3)}$$

Il nous faut donc évaluer les deux limites directionnelles.

```
> Limit(g(x),x=-1,left)=limit(g(x),x=-1,left);
  Limit(g(x),x=-1,right)=limit(g(x),x=-1,right);
```

$$\lim_{x \to (-1)^-} \sqrt[3]{\frac{x\,(x-1)}{x^2-1}} = \infty$$

$$\lim_{x \to (-1)^+} \sqrt[3]{\frac{x\,(x-1)}{x^2-1}} = -\infty \qquad \textbf{(3.4)}$$

Il y a donc une discontinuité infinie en $x = -1$. Ainsi, il y a une asymptote verticale d'équation $x = -1$.

Esquissons le graphique de la fonction g près de cette asymptote verticale. Un tracé sur l'intervalle $[-1,25;\ 0,75]$ devrait faire l'affaire. Contrôlons l'affichage de l'axe des x et de l'axe des y en précisant l'option view=[-3..3,-4..4].

```
> Asymptote_V:=plot([-1,y,y=-4..4],linestyle=3,thickness=0,color=khaki)
  :
  Graphe_2:=plot([x,g(x),x=-1.25..-.75],discont=true):
  display({Graphe_2,Asymptote_V},view=[-3..3,-4..4]);
```



La fonction g est un bel exemple qui nous montre que le calcul de la limite n'est pas nécessairement identique dans $\mathbb{C}$ et dans $\mathbb{R}$. Évaluons donc la limite à droite en $x = -1$ en créant, avec la macro-commande `root`, une fonction f définie par $f(x) = \left(\frac{x\,(x-1)}{x^2-1}\right)^{1/3}$.

```
> f:=x->root(x*(x-1)/(x^2-1),3);
  Limit(f(x),x=-1,right)=limit(f(x),x=-1,right);
```

$$f := x \mapsto root\left(\frac{x\cdot(x-1)}{x^2-1}, 3\right)$$

(3.5)

$$\lim_{x \to (-1)^+} \left(\frac{x(x-1)}{x^2-1} \right)^{1/3} = \left(\mathrm{I}\sqrt{3} + 1 \right) \infty \tag{3.5}$$

C'est convaicant. Cet exemple souligne l'importance d'utiliser la macro-commande `surd` lorsqu'il s'agit d'analyser des fonctions réelles à valeurs réelles.

Pour faire l'étude des asymptotes horizontales, il nous faut évaluer les limites pertinentes à moins l'infini et à plus l'infini.

```
> Limit(g(x),x=-infinity)=limit(g(x),x=-infinity);
  Limit(g(x),x=+infinity)=limit(g(x),x=+infinity)
```

$$\lim_{x \to -\infty} \sqrt[3]{\frac{x(x-1)}{x^2-1}} = 1$$

$$\lim_{x \to \infty} \sqrt[3]{\frac{x(x-1)}{x^2-1}} = 1 \tag{3.6}$$

On a donc une même asymptote horizontale d'équation $y = 1$ dans la partie négative et positive de l'abscisse. Créons le tracé de l'asymptote horizontale et esquissons ensuite le tracé de la fonction g près de cette asymptote.

Pour créer ce tracé, il faut considérer des valeurs pour x de telles sortes que leur image soit assez près de 1 pour que le tracé près de l'asymptote horizontale soit correctement visible sur le graphique. Créons donc séparément deux tracés de la fonction g: un premier sur l'intervalle $[-3;-2]$ dans la partie négative de l'abscisse et un second sur l'intervalle $[2;3]$ dans la partie positive de l'abscisse.

Ces intervalles vont faire l'affaire car ils ont été trouvés après quelques essais. Il est assez rare d'obtenir, du premier coup, les meilleurs intervalles appropriés aux caractéristiques du tracé que l'on cherche à montrer. Attendez-vous qu'il soit souvent nécessaire de faire plusieurs essais.

```
> Asymptote_H:=plot([x,1,x=-3..3],linestyle=3,thickness=0,color=khaki):
  Graphe_3:=plot([x,g(x),x=-3..-2]):
  Graphe_4:=plot([x,g(x),x=2..3]):
  display({Graphe_3,Graphe_4,Asymptote_H},view=[-3..3,-4..4]);
```



Reste maintenant à superposer tous les éléments graphiques obtenus.

```
> display({Trou,Graphe_1,Graphe_2,Graphe_3,Graphe_4,
  Asymptote_V,Asymptote_H},view=[-3..3,-4..4]);
```



Il est nécessaire de poursuivre analytiquement l'étude d'une fonction pour que l'on puisse être en mesure de bien réunir les morceaux que nous venons d'obtenir. C'est seulement après l'introduction de la dérivée et de son interprétation géométrique que nous serons en mesure de réunir correctement ces morceaux. Voici d'ailleurs une surprise si vous croyez que la réunion des morceaux à la droite de l'asymptote verticale se fait toujours en une courbe ouverte vers le bas. Ce n'est pas le cas.

En effet, créez directement le tracé de la fonction g sur l'intervalle $[-3; 3]$. Superposons-y les asymptotes verticale et horizontale ainsi que le "trou". Spécifions l'option `numpoints=200` pour un meilleur rendu de la courbure au voisinage de 0. Finalement, contrôlons l'affichage des axes avec l'option `view=[-3..3,-4..4]`.

```
> Graphe:=plot([x,g(x),x=-3..3],numpoints=200,discont=true):
  display({Trou,Graphe,Asymptote_V,Asymptote_H},view=[-3..3,-4..4]);
```